


THEORY OF COMPILEATION

LECTURE 05

INTERMEDIATE
REPRESENTATION

You are here



Reminder — Previous Steps

```
Potato potato;  
Tomato tomato;  
x = potato + tomato + carrot
```


Lexical analyzer

... <ID,potato> <PLUS> <ID,tomato> <PLUS> <ID,carrot> EOF

Parser

(Annotated)
AST

Symbol
Table



Semantic

symbol	kind	type	properties
x	var	?	
carrot	var	?	
potato	var	Potato	
tomato	var	Tomato	


'carrot' is undefined

'potato' used before initialized

Cannot add 'Potato' and 'Tomato'

Intermediate Representation

- “Neutral” representation between the front-end and the back-end
 - Abstracts away details of the source language
 - Abstract away details of the target language
- In practice, the IR may be biased toward a certain language (*e.g.*, gcc’s GENERIC was created for C)



Intermediate Representation(s)


- Annotated abstract syntax tree
 - Data dependence graph
 - Three address code (3AC)
 - ...
-
- A compiler may have **multiple** intermediate representations and move between them

Three Address Code (3AC)

- Every instruction operates on (at most) three addresses
 - ▶ **result** := **operand1** operator **operand2**
- Close to low-level operations in the machine language
 - ▶ Operator is a basic operation
- Statements in the source language may be mapped to multiple instructions in 3AC

Three Address Code — Example

AST



$a = b * -c + b * -c$

3AC

```
t1 := - c
t2 := b * t1
t3 := - c
t4 := b * t3
t5 := t2 + t4
a := t5
```

Three Address Code

example instructions

instruction	meaning
$x := y \diamond z$	assignment with binary operator
$x := \circ y$	assignment with unary operator
$x := y$	assignment (copy)
$x := \&y$	assign address of y
$x := *y$	assign value in address y (deref)
$*x := y$	assign into address x

(data)

instruction	meaning
goto L	unconditional jump
if $x \triangle y$ goto L	conditional jump

(control)

Array Operations

- Are these 3AC operations?

`x := y[i]`

```
t1 := &y      ; t1 = address-of y
t2 := t1 + i   ; t2 = address of y[i]
x  := *t2      ; load value from y[i]
```

`x[i] := y`

```
t1 := &x      ; t1 = address-of x
t2 := t1 + i   ; t2 = address of x[i]
*t2 := y       ; store value at x[i]
```

Three Address Code — Example

```
int main(void) {  
    int i;  
    int b[10];  
    for (i = 0; i < 10; ++i)  
        b[i] = i*i;  
}
```



```
    i := 0                                ; assignment  
L1:  if i >= 10 goto L2                  ; conditional jump  
    t0 := i * i  
    t1 := &b                            ; t1 = address of b  
    t2 := t1 + i                        ; t2 = address of b[i]  
    *t2 := t0                          ; store i*i at b[i]  
    i := i + 1  
    goto L1  
L2:
```

Creating 3AC

- Assume bottom up parser
 - ▶ Why?
- Creating 3AC via **syntax directed translation**
 - ▶ Semantic attributes:

code	3-address code fragment generated for a nonterminal
var	name of variable that stores the result of code


- ▶ **freshVar** – helper function that returns (the name of) a fresh variable

Creating 3AC: Expressions

production	semantic rule
$S \rightarrow \text{id} := E$	$S.\text{code} = E.\text{code} \parallel (\text{id}.\text{name} \text{ ':=' } E.\text{var})$
$E \rightarrow E_1 + E_2$	$E.\text{var} = \text{freshVar}();$ $E.\text{code} = E_1.\text{code} \parallel E_2.\text{code} \parallel (E.\text{var} \text{ ':=' } E_1.\text{var} \text{ '+' } E_2.\text{var})$
$E \rightarrow E_1 * E_2$	$E.\text{var} = \text{freshVar}();$ $E.\text{code} = E_1.\text{code} \parallel E_2.\text{code} \parallel (E.\text{var} \text{ ':=' } E_1.\text{var} \text{ '*' } E_2.\text{var})$
$E \rightarrow - E_1$	$E.\text{var} = \text{freshVar}();$ $E.\text{code} = E_1.\text{code} \parallel (E.\text{var} \text{ ':=' '-' } E_1.\text{var})$
$E \rightarrow (E_1)$	$E.\text{var} = E_1.\text{var};$ $E.\text{code} = E_1.\text{code}$
$E \rightarrow \text{id}$	$E.\text{var} = \text{id}.\text{name};$ $E.\text{code} = ''$

(we use \parallel to denote concatenation of intermediate code fragments)

Example



Creating 3AC

- Option 1
accumulate code in AST attributes
(which is what we just did)
- Option 2 (incremental translation)
emit IR code to a file during compilation
 - ▶ Possible when for every production, the code of the left-hand side is constructed from a concatenation of the code of the right-hand side **in some fixed order**

Expressions and Assignments

This is me cheating 😬

production	semantic action
$S \rightarrow \text{id} := E$	emit (id.name $:=$ E.var)
$E \rightarrow E_1 \diamond E_2$	E.var := freshVar (); emit (E.var $:=$ $E_1.\text{var} \diamond E_2.\text{var}$)
$E \rightarrow - E_1$	E.var := freshVar (); emit (E.var $:=$ $- E_1.\text{var}$)
$E \rightarrow (E_1)$	E.var := $E_1.\text{var}$
$E \rightarrow \text{id}$	E.var := id.name

$\diamond \in \{ '+', '*' \}$

Creating 3AC: Control Statements

- 3AC only supports jumps (conditional and unconditional)
 - ▶ Add labels to code
 - ▶ Store labels in attributes:

instruction	meaning
<code>goto L</code>	unconditional jump
<code>if x Δ y goto L</code>	conditional jump

<code>begin</code>	marks beginning of code fragment
<code>next</code>	follows end of code fragment

- `freshLabel` – helper function that allocates a new, unused label

Creating 3AC: Control statements

For all simple (non-control) statements —

$x := x + 2 * y$

$l_1:$ $t_1 := 2 * y$
 $x := x + t_1$
 $l_2:$

S.begin:

S.code


S.next:

production	semantic action
$S \rightarrow id := E$	$S.begin = \text{freshLabel}();$ $S.next = \text{freshLabel}();$ $S.code = (S.begin ':' $ $E.code id.name ':=' E.var) $ $(S.next ':')$


Creating 3AC: Control statements

$S \rightarrow \text{if } E \text{ then } S_1$

if 1 then x := x + 1



S.begin:




production	semantic action
$S \rightarrow \text{if } E \text{ then } S_1$	$S.\text{begin} = \text{freshLabel}(); \quad S.\text{next} = \text{freshLabel}();$ $S.\text{code} = (S.\text{begin} ':') E.\text{code} $ $(\text{'if' } E.\text{var} '=' '0' \text{'goto' } S.\text{next}) $ $S_1.\text{code} (S.\text{next} ':')$

Creating 3AC: Control statements

$S \rightarrow \text{if } E \text{ then } S_1$
 $\quad \text{else } S_2$


if 1 then x:=x+1 else x:=x-1



this line is
 both S_{next}
 and $S_2.\text{next}$



$S.\text{begin}:$




production	semantic action
$S \rightarrow \text{if } E \text{ then } S_1$ $\quad \text{else } S_2$	$S.\text{begin} = \text{freshLabel}(); \quad S.\text{next} = \text{freshLabel}();$ $S.\text{code} = (S.\text{begin} ':') \parallel E.\text{code} \parallel$ $(\text{'if' } E.\text{var} \text{'=' '0' 'goto' } S_2.\text{begin}) \parallel S_1.\text{code} \parallel$ $(\text{'goto' } S.\text{next}) \parallel S_2.\text{code} \parallel (S.\text{next} ':')$


Creating 3AC: Control statements

$S \rightarrow \text{while } E \text{ do } S_1$

while 1 do x := x + 1



S.begin:



production	semantic action
$S \rightarrow \text{while } E \text{ do } S_1$	$S.\text{begin} = \text{freshLabel}(); \quad S.\text{next} = \text{freshLabel}();$ $S.\text{code} = (S.\text{begin} ':') \parallel E.\text{code} \parallel$ $(\text{'if' } E.\text{var} '=' '0' \text{'goto' } S.\text{next}) \parallel$ $S_1.\text{code} \parallel (\text{'goto' } S.\text{begin}) \parallel (S.\text{next} ':')$


Boolean Expressions

This is me cheating again 😬

production	semantic action
$E \rightarrow E_1 \circledast E_2$	$E.var := \text{freshVar}(); E.code := (E.var \text{ ':=' } E_1.var \circledast E_2.var)$
$E \rightarrow \text{not } E_1$	$E.var := \text{freshVar}(); E.code := (E.var \text{ ':=' } \text{'not'} E_1.var)$
$E \rightarrow (E_1)$	$E.var := E_1.var$
$E \rightarrow \text{true}$	$E.var := \text{freshVar}(); E.code := (E.var \text{ ':=' } \text{'1'})$
$E \rightarrow \text{false}$	$E.var := \text{freshVar}(); E.code := (E.var \text{ ':=' } \text{'0'})$

$\circledast \in \{\text{'and'}, \text{'or'}\}$

Boolean expressions via jumps

production	semantic action
$E \rightarrow id_1 \ominus id_2$ 	$E.var := \text{freshVar}();$ $t := \text{freshLabel}(); \text{after} := \text{freshLabel}();$ $E.code := (\text{'if' } id_1.var \ominus id_2.var \text{'goto' } t) \parallel$ $(E.var \text{' := ' '0'}) \parallel$ $(\text{'goto' } after) \parallel$ $(t \text{' : '}) \parallel (E.var \text{' := ' '1'}) \parallel (after \text{' : '})$

$\ominus \in \{ '<', '<=', '>', '>=' \}$

$a < b$

if $a < b$ **goto** l_1

$t_1 := 0$


goto l_2

$l_1:$ $t_1 := 1$

$l_2:$

Example

$a < b \text{ or } (c < d \text{ and } e < f)$



Short-Circuit Evaluation

- Second argument of a Boolean operator is *only evaluated* if the first argument does not already determine the outcome

$(x \text{ and } y)$ *is equivalent to* `if x then y else false`

$(x \text{ or } y)$ *is equivalent to* `if x then true else y`

Short-Circuit Evaluation

$a < b$ **or** $(c < d$ **and** $e < f)$

100: if $a < b$ goto 103

101: $T_1 := 0$

102: goto 104

103: $T_1 := 1$

104: if $c < d$ goto 107

105: $T_2 := 0$

106: goto 108

107: $T_2 := 1$

108: if $e < f$ goto 111

109: $T_3 := 0$

110: goto 112

111: $T_3 := 1$

112: $T_4 := T_2$ **and** T_3

113: $T_5 := T_1$ **or** T_4

naïve evaluation

100: if $a < b$ goto 106

101: if $c < d$ goto 103

102: goto 104

103: if $e < f$ goto 106

104: $T := 0$

105: goto 107

106: $T := 1$

107:

short-circuit evaluation

Boolean Expressions & Control Structures

$S \rightarrow$ `if B then S1`
| `if B then S1 else S2`
| `while B do S1`

- For every Boolean expression B, we attach two attributes

B.falseLabel	target label for a jump when condition B evaluates to false
---------------------	--


B.trueLabel	target label for a jump when condition B evaluates to true
--------------------	---

- For every statement S we attach an attribute


S.next	the label of the next code to execute after S
---------------	--

Boolean Expressions

production	semantic action
$B \rightarrow \text{true}$	$B.\text{code} = \text{'goto' } B.\text{trueLabel}$
$B \rightarrow \text{false}$	$B.\text{code} = \text{'goto' } B.\text{falseLabel}$
$B \rightarrow id_1 \ominus id_2$	$B.\text{code} = (\text{'if' } id_1.\text{var} \ominus id_2.\text{var} \text{'goto' } B.\text{trueLabel}) \parallel (\text{'goto' } B.\text{falseLabel});$
$B \rightarrow B_1 \text{ or } B_2$	$B_1.\text{trueLabel} = B.\text{trueLabel};$ $B_1.\text{falseLabel} = \text{freshLabel}();$ $B_2.\text{trueLabel} = B.\text{trueLabel};$ $B_2.\text{falseLabel} = B.\text{falseLabel};$ $B.\text{code} = B_1.\text{code} \parallel (B_1.\text{falseLabel} \text{'.'}) \parallel B_2.\text{code}$



Boolean Expressions



```

B1.trueLabel = B.trueLabel;
B1.falseLabel = freshLabel();
B2.trueLabel = B.trueLabel;
B2.falseLabel = B.falseLabel;
B.code = B1.code || (B1.falseLabel ':') || B2.code
    
```


$x < a \text{ or } x > b$

```

if x < a goto l1
goto l3
l3: if x > b goto l1
      goto l2
l1: ...
l2: ... (later on)
    
```

B.code

$l_3 = B_1.\text{falseLabel}$



Boolean Expressions



production	semantic action
$B \rightarrow B_1 \text{ or } B_2$	$B_1.\text{trueLabel} = B.\text{trueLabel};$ $B_1.\text{falseLabel} = \text{freshLabel}();$ $B_2.\text{trueLabel} = B.\text{trueLabel};$ $B_2.\text{falseLabel} = B.\text{falseLabel};$ $B.\text{code} = B_1.\text{code} \parallel (B_1.\text{falseLabel} ':') \parallel B_2.\text{code}$
$B \rightarrow B_1 \text{ and } B_2$	$B_1.\text{trueLabel} = \text{freshLabel}();$ $B_1.\text{falseLabel} = B.\text{falseLabel};$ $B_2.\text{trueLabel} = B.\text{trueLabel};$ $B_2.\text{falseLabel} = B.\text{falseLabel};$ $B.\text{code} = B_1.\text{code} \parallel (B_1.\text{trueLabel} ':') \parallel B_2.\text{code}$
$B \rightarrow \text{not } B_1$	$B_1.\text{trueLabel} = B.\text{falseLabel};$ $B_1.\text{falseLabel} = B.\text{trueLabel};$ $B.\text{code} = B_1.\text{code};$
$B \rightarrow (B_1)$	$B_1.\text{trueLabel} = B.\text{trueLabel}; B_1.\text{falseLabel} = B.\text{falseLabel}; B.\text{code} = B_1.\text{code};$
$B \rightarrow \text{id}_1 \ominus \text{id}_2$	$B.\text{code} = ('if' \text{id}_1.\text{var} \ominus \text{id}_2.\text{var} \text{'goto'} B.\text{trueLabel}) \parallel ('goto' B.\text{falseLabel});$
$B \rightarrow \text{true}$	$B.\text{code} = \text{'goto'} B.\text{trueLabel}$
$B \rightarrow \text{false}$	$B.\text{code} = \text{'goto'} B.\text{falseLabel}$

Control Structures: next

Note. We are **changing** the behavior of 'next' from before; 'begin' is no longer needed.

production	semantic action
$P \rightarrow S$	$S.next = \text{freshLabel}();$ $P.code = S.code \parallel (S.next ':')$
$S \rightarrow S_1 ; S_2$	$S_1.next = \text{freshLabel}();$ $S_2.next = S.next;$ $S.code = S_1.code \parallel (S_1.next ':') \parallel S_2.code$


$S_1 ; S_2 ; S_3$



Control Structures: conditional


production	semantic action
$S \rightarrow \text{if } B \text{ then } S_1$	$B.\text{trueLabel} = \text{freshLabel}();$ $B.\text{falseLabel} = S.\text{next};$ $S_1.\text{next} = S.\text{next};$ $S.\text{code} = B.\text{code} \parallel (B.\text{trueLabel} ':') \parallel S_1.\text{code}$

if $a < b$ **then** $x := x + 1$



$B.\text{trueLabel}:$


$B.\text{falseLabel}:$
 (= $S.\text{next}$)



Control Structures: conditional

production	semantic action
$S \rightarrow \text{if } B \text{ then } S_1 \text{ else } S_2$	$B.\text{trueLabel} = \text{freshLabel}();$ $B.\text{falseLabel} = \text{freshLabel}();$ $S_1.\text{next} = S.\text{next};$ $S_2.\text{next} = S.\text{next};$ $S.\text{code} =$ $B.\text{code} \parallel (B.\text{trueLabel} ':') \parallel S_1.\text{code} \parallel ('goto' S.\text{next})$ $\parallel (B.\text{falseLabel} ':') \parallel S_2.\text{code}$


if $a < b$ **then** $x := x + 1$ **else** $x := x - 1$




$B.\text{trueLabel}:$

$B.\text{falseLabel}:$

$S.\text{next}:$



Boolean Expressions



```
B.trueLabel = freshLabel();
```

```
B.falseLabel = S.next;
```

```
S1.next = S.next;
```

```
S.code = B.code ||
```

```
(B.trueLabel ':') || S1.code
```


```
if x < a or x > b then S1
```

```

if x < a goto l7
goto l8
l8: if x > b goto l7
    goto l6
l7: ... S1.code ...
l6: ...
    
```

B.code


$l_8 = B_1.\text{falseLabel}$



$B.\text{trueLabel} = l_7$

$B.\text{falseLabel} = l_6$

Boolean Expressions – S-Attributed



$S.code = B.code$

$|| (B.trueLabel ':') || S_1.code$
 $|| (B.falseLabel ':')$

if $x < a$ **or** $x > b$ **then** S_1

```

if  $x < a$  goto  $l_6$ 
goto  $l_7$ 
 $l_7$ : if  $x > b$  goto  $l_8$ 
      goto  $l_9$ 
 $l_6$ : goto  $l_8$ 
 $l_8$ : ...  $S_1.code$  ...
 $l_9$ : ...
    
```

$B \rightarrow id_1 \ominus id_2$ $B.trueLabel = freshLabel(); B.falseLabel = freshLabel();$
 $B.code = ('if' id_1.var \ominus id_2.var 'goto' B.trueLabel) ||$
 $('goto' B.falseLabel)$

$B \rightarrow B_1 \text{ or } B_2$ $B.trueLabel = B_2.trueLabel;$
 $B.falseLabel = B_2.falseLabel;$
 $B.code = B_1.code || (B_1.falseLabel ':') || B_2.code$
 $|| (B_1.trueLabel ':') || ('goto' B.trueLabel)$

Boolean Expressions – S-Attributed

production	semantic action
$B \rightarrow \text{true}$	$B.\text{trueLabel} = \text{freshLabel}(); B.\text{falseLabel} = \text{freshLabel}();$ $B.\text{code} = \text{'goto' } B.\text{trueLabel}$
$B \rightarrow \text{false}$	$B.\text{trueLabel} = \text{freshLabel}(); B.\text{falseLabel} = \text{freshLabel}();$ $B.\text{code} = \text{'goto' } B.\text{falseLabel}$
$B \rightarrow \text{id}_1 \ominus \text{id}_2$	$B.\text{trueLabel} = \text{freshLabel}(); B.\text{falseLabel} = \text{freshLabel}();$ $B.\text{code} = (\text{'if' id}_1.\text{var } \ominus \text{id}_2.\text{var 'goto' } B.\text{trueLabel}) \parallel (\text{'goto' } B.\text{falseLabel})$
$B \rightarrow B_1 \text{ or } B_2$	$B.\text{trueLabel} = B_2.\text{trueLabel};$ $B.\text{falseLabel} = B_2.\text{falseLabel};$ $B.\text{code} = B_1.\text{code} \parallel (B_1.\text{falseLabel ':'}) \parallel B_2.\text{code} \parallel$ $(B_1.\text{trueLabel ':'}) \parallel (\text{'goto' } B.\text{trueLabel});$
$S \rightarrow \text{if } B \text{ then } S_1$	$S.\text{code} = B.\text{code} \parallel (B.\text{trueLabel ':'}) \parallel S_1.\text{code} \parallel (B.\text{falseLabel ':'})$
$S \rightarrow \text{if } B \text{ then } S_1$ else S_2	$S.\text{next} = \text{freshLabel}();$ $S.\text{code} = B.\text{code} \parallel (B.\text{trueLabel ':'}) \parallel S_1.\text{code} \parallel (\text{'goto' } S.\text{next})$ $(B.\text{falseLabel ':'}) \parallel S_2.\text{code} \parallel (S.\text{next ':'})$

Boolean Expressions – S-Attributed

with emit(..)

production	semantic action
$B \rightarrow id_1 \ominus id_2$	$B.trueLabel = freshLabel(); B.falseLabel = freshLabel();$ $B.code = ('if' id_1.var \ominus id_2.var 'goto' B.trueLabel) ('goto' B.falseLabel)$
$S \rightarrow if\ B\ then\ S_1$	$S.code = B.code (B.trueLabel ':') S_1.code (B.falseLabel ':')$



production	semantic action
$B \rightarrow id_1 \ominus id_2$	$B.trueLabel = freshLabel(); B.falseLabel = freshLabel();$ $emit('if' id_1.var \ominus id_2.var 'goto' B.trueLabel 'goto' B.falseLabel)$
$S \rightarrow if\ B\ M\ then\ S_1$	$emit(B.falseLabel ':')$
$M \rightarrow \epsilon$	$emit(B.trueLabel ':')$

LLVM IR

- LLVM = Low-Level Virtual Machine
 - ▶ A well-known misnomer: it is (mostly) a **compiler** framework
- Flat IR, similar to 3-address code in nature
- All values are ***typed*** (unlike assembly)

`%num = add i32 %inp, 48`

target opcode type operands

LLVM IR — Hello World

global
symbol

external
symbol

start
function

start
block

end
block

end
function

```
@str = internal constant [14 x i8] c"hello, world\0A\00"  
  
declare i32 @printf(i8*, ...)  
  
define i32 @main() {  
  entry:  
    %tmp1 = getelementptr [14 x i8], [14 x i8]* @str, i32 0, i32 0  
    %tmp2 = call i32 (i8*, ...) @printf( i8* %tmp1 )  
  
    ret i32 42  
}  

```

type annotations

local
symbol

LLVM IR — Types

- Numeric

- ▶ Integers — any bit width!

- ▶ Floating point: 16, 32, 64 bit

i1
i32
i1942652

half, float, double

- Pointer

i8 *

- Label (= code address)

label

- Aggregate

- ▶ Array (fixed dimensions)

[40 x i32]
[12 x [10 x float]]

- ▶ Struct

{ float, [4 x i32] }

LLVM IR — Memory

```
%a = alloca i32
store i32 5, i32* %a
%rd = load i32, i32* %a
```

allocate on stack
return type is a pointer

write to memory address $*a = 5$

read from memory address $rd = *a$

```
%a = alloca [4 x i32]
%el = getelementptr [4 x i32],
    [4 x i32]* %a, i32 0, i32 1
store i32 5, i32* %el
%rd = load i32, i32* %el
```

aggregate type

compute address of
element in aggregate
return type is a pointer

base pointer

idx₁

idx₂

LLVM IR — Memory

getelementptr



All



Maps



Videos



News



Images



More

Settings

Tools

About 33,900 results (0.30 seconds)

The Often Misunderstood GEP Instruction — LLVM 10 ...

[https://llvm.org › docs › GetElementPtr](https://llvm.org/docs/GetElementPtr) ▼

This document seeks to dispel the mystery and confusion surrounding LLVM's **GetElementPtr** (GEP) instruction. Questions about the wily GEP instruction are ...

LLVM — Gotchas

- LLVM blocks must start with a label and end with a *Terminator instruction*

ret br switch indirectbr invoke callbr resume
catchswitch catchret cleanupret unreachable



- LLVM variables are *Single Static Assignment* (SSA)

► Wait, what?

Single Static Assignment

- Variables can only appear on the left-hand side of *one* assignment.

```
cond:
    %b = icmp slt i32 %i, %j
    br i1 %b, label %then,
        label %else
```

```
then:
    %max = or i32 0, %j
    br label %exit
```

```
else:
    %max = or i32 0, %i
    br label %exit
```

```
exit:
    ret i32 %max
```



```
cond:
    %b = icmp slt i32 %i, %j
    br i1 %b, label %then,
        label %else
```

```
then:
    %max1 = or i32 0, %j
    br label %exit
```

```
else:
    %max2 = or i32 0, %i
    br label %exit
```

```
exit:
    %max = phi i32 [ %max1, %then ],
                [ %max2, %else ]
    ret i32 %max
```



Coming Up

